
1 Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 9.1.4.1 which is part of the 5G NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Shaun Harry


Shaun.harry@keysight.com
1.1 Verification Test Summary 
Test Case: 
9.1.4.1
ATS Version:
iwd-TTCN3-B2019-06_D20wk11
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
HiSilicon Balong 5000
Verification Status:
PASS
2 Corrections required
2.1 Change 1

	Function name
	function f_TC_9_1_4_1_NR5GC

	Reason for change
	UE is in IDLE state at the end of test body..

	Summary of change
	Use STATE_IDLE_1A for postamble.

	TTCN module
	ConfigurationUpdate_NR5GC

	MCC160 Comment
	Accepted


Before 

function f_TC_9_1_4_1_NR5GC() runs on NR5GC_PTC

  { // Generic UE configuration update / New 5G-GUTI, NITZ, registration requested, Network slicing indication, New Allowed NSSAI /

    // acknowledgement from the UE

    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_CellsOnDifferentPLMN);

    var NR_CellId_Type v_NGC_NASCellG := f_NR5GC_MapNASCell(ngc_CellG, NG_CellsOnDifferentPLMN);

    f_NR5GC_Init_NAS(NR_4, NG_CellsOnDifferentPLMN);

    //Create and configure cell

    f_NR_CellConfig_Def(v_NGC_NASCellA);

    f_NR_CellConfig_Def(v_NGC_NASCellG);

    f_NR5GC_Preamble (v_NGC_NASCellA, STATE_CONNECTED_3A);

    f_NR_TestBody_Set(true);

    fl_TC_9_1_4_1_TestBody();

    f_NR_TestBody_Set(false);

    // Finish

    f_NR_Postamble(v_NGC_NASCellG, STATE_CONNECTED_3A);

  }

After change

function f_TC_9_1_4_1_NR5GC() runs on NR5GC_PTC

  { // Generic UE configuration update / New 5G-GUTI, NITZ, registration requested, Network slicing indication, New Allowed NSSAI /

    // acknowledgement from the UE

    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_CellsOnDifferentPLMN);

    var NR_CellId_Type v_NGC_NASCellG := f_NR5GC_MapNASCell(ngc_CellG, NG_CellsOnDifferentPLMN);

    f_NR5GC_Init_NAS(NR_4, NG_CellsOnDifferentPLMN);

    //Create and configure cell

    f_NR_CellConfig_Def(v_NGC_NASCellA);

    f_NR_CellConfig_Def(v_NGC_NASCellG);

    f_NR5GC_Preamble (v_NGC_NASCellA, STATE_CONNECTED_3A);

    f_NR_TestBody_Set(true);

    fl_TC_9_1_4_1_TestBody();

    f_NR_TestBody_Set(false);

    // Finish

    f_NR_Postamble(v_NGC_NASCellG, STATE_IDLE_1A);

  }
2.2 Change 2

	Function name
	fl_TC_9_1_4_1_TestBody

	Reason for change
	1. SRB2 has been established in preamble so configuration update procedure should be on SRB2.
2. Need to update new GUTI in cell info so that Paging functions correctly at Step 4.
3. The UE will issue one +CTZE indication on receiving CONFIGURATION UPDATE COMMAND at step 12. So Time zone information should be checked before querying operator name.
4. In order to deliver the CTZE response to TTCN, need to add a dummy AT command before Step 12.
5. The hour info including in CTZE response should be 13, instead of 14.
6. RRC connection should be released after steb 32A
7. The SST in s_nssai list in Step 31 and step 64 should be corrected. SST in S-NSSAI value 2 should be ‘00000010’B and SST in S-NSSAI value 3 should be ‘00000011’B.
8. In Step33, the original implementation can’t get the requestedNSSAI. requestedNSSAI in Registration Request is included in the PiggybackedPduList.
9. At step 50-52, SRB2 will be established in the case of initial registration but not established for mobility rregtistration. So configuration update signaling at steps 66-67 can use a different SRB depending on the type of registration performed. 

	Summary of change
	1. Use SRB2 for CONFIGURATION UPDATE COMMAND / COMPLETE throughout.
2. Update the GUTI in cell info before Step 4.
3. Move the checking of CTZE before the checking of operator name.
4. Add trigger to deliver the CTZE response to TTCN.
5. Correct the hour checking of CTZE.
6. Add step 32B to release the RRC connection.
7. Correct the SST value used in Step 31 and Step 64.
8. Correct the way to get requestedNSSAI in Step 33.
9. Choose SRB depending on registratrion type at steps 66-67
Note : A prose CR on 38.523-1 associated with changes 6, 7 and 9 will be raised at RAN5#87

	TTCN module
	ConfigurationUpdate_NR5GC

	MCC160 Comment
	1, 2, 3, 5, 8, 9. Accepted
4 Accepted but also added f_UT_AT_SubscribeTimeZoneReporting to turn on timezone reporting 

6, 7. Accepted subject to prose CR approval


Before 

function fl_TC_9_1_4_1_TestBody() runs on NR5GC_PTC

  {

    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_CellsOnDifferentPLMN);

    var NR_CellId_Type v_NGC_NASCellG := f_NR5GC_MapNASCell(ngc_CellG, NG_CellsOnDifferentPLMN);

    var NG_NAS_GutiParameters_Type v_GutiParamsCellA := f_NR5GC_CellInfo_GetGutiParameters(v_NGC_NASCellA);

    var NG_NAS_GutiParameters_Type v_NewGutiParams := v_GutiParamsCellA;

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var NG_5G_S_TMSI v_S_Tmsi :=  f_NR_Get5G_S_TMSI_FromGuti (v_GutiParamsCellA);

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;

    var NAS_PlmnId v_PLMN_CellA := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA));

    var NAS_PlmnId v_PLMN_CellG := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellG));

    var charstring v_Quote := oct2char ('22'O);

    var charstring v_FullNameString := "FullName12345678";

    var charstring v_ShortNameString := "SName123"; // @sic R5-198186 sic@

    var charstring v_Expression_CTZR := "*"&v_Quote&"(*)"&v_Quote&",([0-9])*"&v_Quote&"(*)"&v_Quote&"*"; // used for LTZ, DST and Time checking

    var charstring v_AT_Response;

    var PCCH_Config v_PcchConfig;

    var NSSAI v_RequestedNSSAI;

    var TypeOfRegistration_Type v_TypeOfRegistration;

    var integer i;

    var template (value) NSSAI v_ConfiguredNSSAI := {

      iei := '31'O,

      iel := '04'O,

      s_nssai := {cs_S_NSSAI(omit, '01'O, '01'O, omit), cs_S_NSSAI(omit, '01'O, '02'O, omit)}

      };

    var template (value) NSSAI v_AllowedNSSAI:= {

      iei := '15'O,

      iel := '06'O,

      s_nssai := {cs_S_NSSAI(omit, '01'O, '01'O, omit), cs_S_NSSAI(omit, '01'O, '02'O, omit), cs_S_NSSAI(omit, '01'O, '03'O, omit)}

      };

    timer t_T3555 := f_NR_SetTimerToleranceMin(v_NGC_NASCellA, nasTimer, 6.0);

    //@siclog "Step 1" siclog@

    v_S_Tmsi := f_CreateNewS_TMSING_5G_S_TMSI(v_S_Tmsi);

    v_NewGutiParams.AMF_SetId := substr(v_S_Tmsi, 0, 10);

    v_NewGutiParams.AMF_Pointer := substr(v_S_Tmsi, 10, 6);

    v_NewGutiParams.NG_TMSI := substr(v_S_Tmsi, 16, 32);

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(cs_ConfigUpdateInd('0'B, '1'B),

                                                                      f_NG_GutiParameters2MobileIdentity(v_PLMN_CellA, v_NewGutiParams)))));

    //@siclog "Step 2" siclog@

    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    //@siclog "Step 3" siclog@

    f_NR_RRCRelease(v_NGC_NASCellA);

    v_PcchConfig := f_NR_GetSIB1_PcchConfig(v_NGC_NASCellA);

    //@siclog "Step 4" siclog@

    // Page with the S-TMSI used in the new GUTI

    f_NR_Paging(v_NGC_NASCellA,

                cs_PagingRecordList_OneRecord(cs_PagingUE_Identity_5G_S_TMSI(v_S_Tmsi)),

                v_PcchConfig.defaultPagingCycle); //f_NR_GetSIB1_PcchConfig(v_NGC_NASCellA).defaultPagingCycle;

    //@siclog "Steps 5-11" siclog@

    v_ReceivedMsg := f_NR5GC_RRCIdleToConnected_Steps2_4(v_NGC_NASCellA);

    f_NR5GC_RRC_ConnectedState3N_Steps5_8(v_NGC_NASCellA, v_ReceivedMsg);

    //@siclog "Step 12" siclog@

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(-, -, -, -, -,

                                                                      cs_NetworkName ('43'O, v_FullNameString),

                                                                      cs_NetworkName ('45'O, v_ShortNameString), // @sic R5-198186 sic@

                                                                      cs_TimeZone('40'O), // @sic R5-198186 sic@

                                                                      cs_TimeZoneAndTime('01'O, '40'O, '21'O, '31'O, '83'O, '25'O,'40'O),

                                                                      cs_DaylightSavingTime_tlv('01'B))))); // @sic R5-198186 sic@

    //@siclog "Steps 13" siclog@

    t_T3555.start;

    alt{

        [] SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)))

            {

                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");

            }

        [] t_T3555.timeout {}

    }

    //@siclog "Steps 14" siclog@

    v_AT_Response := f_UT_ChkOperatorName (UT, LONG_NAME);

    f_AT_ResponseCheck (v_AT_Response,

                        v_FullNameString,

                        "Step 14 (long network name)");

    v_AT_Response := f_UT_ChkOperatorName (UT, SHORT_NAME);

    f_AT_ResponseCheck (v_AT_Response,

                        v_ShortNameString,

                        "Step 14 (short network name)");

    v_AT_Response := f_UT_ChkTimeZone (UT);

    f_AT_ResponseCheck(v_AT_Response,

                       "*[0-9]#(2)/[0-9]#(2)/[0-9]#(2),[0-9]#(2):[0-9]#(2):[0-9]#(2)\+04*", // only TZ to be checked

                       "Step 14 (local time zone)");

    v_AT_Response := f_UT_AT_CheckTimezoneReport (UT); // CTZE response expected

    f_AT_ResponseCheck(v_AT_Response,

                       "2010/04/12,14:[0-9]#(2):[0-9]#(2)", // TZ GMT+1, dst (1 char) ignored, check date and hour, ignore mins, secs

                       "Step 14 (UT and local time zone)",

                       v_Expression_CTZR,

                       2);

    f_AT_ResponseCheck(v_AT_Response,

                       "1", // DST=1

                       "Step 14  DST",

                       v_Expression_CTZR,

                       1);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    //@siclog "Steps 15" siclog@

    // Switch off the UE

    f_NR_UE_DeRegisterOnSwitchOff (v_NGC_NASCellA, STATE_CONNECTED_3A);

    //@siclog "Step 16" siclog@

    // Switch on UE

    f_UT_SwitchOnUE(UT);

    //@siclog "Steps 17-19" siclog@

    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_Secure)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    //@siclog "Steps 20-30" siclog@

    f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo,

                                v_NGC_NASCellA, v_ReceivedMsg,

                                v_PLMN_CellA,

                                TESTMode_OFF,

                                Initial_Secure);

    //@siclog "Step 31" siclog@

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,

                                                   f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellA),

                                                   cs_NG_TAIListNonConsecutive(v_PLMN_CellA,

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(v_NGC_NASCellA))}),

                                                   -, cs_NSSAI_SST1eMBB ('15'O), // default value

                                                   -, v_ConfiguredNSSAI);

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Step 31A - 31E" siclog@

    f_NR5GC_RRC_Idle_Steps15_20(v_NGC_NASCellA,TESTMode_OFF,noRrcConnectionRelease); // @sic R5-198186 sic@

    //@siclog "Step 32A" siclog@ // @sic R5-198186 sic@

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(cs_ConfigUpdateInd('1'B, '1'B)))));

    //@siclog "Step 32A" siclog@

    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    //SS releases the RRC connection. Not in prose FFS!

    //@siclog "Step ?" siclog@

    //f_NR_RRCRelease(v_NGC_NASCellA);
    //@siclog "Step 33" siclog@

    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility, -, -, -, -, -, -, ?)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 33");

    }

    // Need to check received contents (cannot hard-code a template to test this on a record of type)

    v_RequestedNSSAI.s_nssai := v_ReceivedMsg.Pdu.Msg.registration_Request.requestedNSSAI.s_nssai;

    for (i := 0; i < lengthof(v_RequestedNSSAI.s_nssai); i := i+1) {

      if (match (v_RequestedNSSAI.s_nssai[i].sd, v_ConfiguredNSSAI.s_nssai[0].sd) or match (v_RequestedNSSAI.s_nssai[i].sd, v_ConfiguredNSSAI.s_nssai[1].sd)) {

        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 33");

      } else {

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unknown Requested NSSAI");

      }

    }

    //@siclog "Step 34 - 42" siclog@

    // Void // @sic R5-198186 sic@

    //@siclog "Step 43" siclog@

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellA),

                                                   cs_NG_TAIListNonConsecutive(v_PLMN_CellA,

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(v_NGC_NASCellA))}),

                                                   -, v_AllowedNSSAI,

                                                   -,-,-,

                                                   v_GMM_MobilityInfo.SessionStatus //@sic R5-198509 sic@

                                                   );

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Step 44" siclog@

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

    }

    //@siclog "Step 44A" siclog@

    f_NR_RRCRelease(v_NGC_NASCellA); // @sic R5-198186 sic@

    //@siclog "Step 45" siclog@

    v_AT_Response := f_UT_ReadAllowedNSSAI(UT, f_ConvertPLMNtoString(v_PLMN_CellA));

    if (match (v_AT_Response, f_Convert_ListOfNSSAI_ForATCommand(v_AllowedNSSAI))) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 45");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 45");

    }

    //@siclog "Step 45A - 45H" siclog@

    // The test steps 1 to 8 of generic test procedure in TS 38.508-1 Table 4.5.4.2-3 are performed // @sic R5-198186 sic@

    f_NR5GC_RRC_ConnectedState3N_Def (v_NGC_NASCellA);

    //@siclog "Step 46" siclog@

    v_AllowedNSSAI.s_nssai := {cs_S_NSSAI(omit, '01'O, '01'O, omit), cs_S_NSSAI(omit, '01'O, '02'O, omit)};

    v_AllowedNSSAI.iel := '04'O;

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(cs_ConfigUpdateInd('0'B, '1'B), -, -, v_AllowedNSSAI))));

    //@siclog "Step 47" siclog@

    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    //@siclog "Step 48" siclog@

    v_AT_Response := f_UT_ReadAllowedNSSAI(UT, f_ConvertPLMNtoString(v_PLMN_CellA));

    if (match (v_AT_Response, f_Convert_ListOfNSSAI_ForATCommand(v_AllowedNSSAI))) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 48");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 48");

    }

    //SS releases the RRC connection.

    //@siclog "Step 48A" siclog@

    f_NR_RRCRelease(v_NGC_NASCellA);

    //@siclog "Step 49" siclog@

    f_NR_SetCellPower (v_NGC_NASCellG, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    f_NR_SetCellPower (v_NGC_NASCellA, tsc_NR_NonSuitableCellSSS_EPRE_FR1, tsc_NR_NonSuitableCellSSS_EPRE_FR2);

    //@siclog "Steps 50-52" siclog@

    v_TypeOfRegistration := f_NR5GC_InitialOrMobilityRegistrationRequest (v_GMM_MobilityInfo, v_ReceivedMsg, v_NGC_NASCellG);

    if (v_TypeOfRegistration == Initial_Secure) {

      // @siclog "Steps 53a1 - 53a10" siclog@

      f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo, v_NGC_NASCellG, v_ReceivedMsg, v_PLMN_CellG, TESTMode_OFF, v_TypeOfRegistration);

    }

    //@siclog "Step 53b" siclog@

    // do nothing if mobility

    //@siclog "Step 64" siclog@

    v_ConfiguredNSSAI.s_nssai := {cs_S_NSSAI(omit, '01'O, '01'O, omit), cs_S_NSSAI(omit, '01'O, '02'O, omit)};

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellG),

                                                   cs_NG_TAIListNonConsecutive(v_PLMN_CellG,

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(v_NGC_NASCellG))}),

                                                   -, cs_NSSAI_SST1eMBB ('15'O), // default value

                                                   -, v_ConfiguredNSSAI,

                                                   -,

                                                   v_GMM_MobilityInfo.SessionStatus //@sic R5-198509 sic@

                                                   );

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellG,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    if (v_TypeOfRegistration == Initial_Secure) {

      // @siclog "Steps 65a1 - 65a5a1" siclog@

      f_NR5GC_RRC_Idle_Steps15_20 (v_NGC_NASCellG, TESTMode_OFF, noRrcConnectionRelease);

    } else {

      //@siclog "Step 65b" siclog@

      SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellG, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

      if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

      }

    }

    //@siclog "Step 66" siclog@

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellG,

                                   tsc_NR_RbId_SRB1,
                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND( cs_ConfigUpdateInd('0'B, '1'B), // @sic R5-198186 sic@

                                                                       -, -, -, -, -, -, -, -, -, -, -, cs_NetworkSlicingInd('0'B, '1'B)))));

    //@siclog "Step 67" siclog@

    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellG, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    //@siclog "Step 67A" siclog@

    f_NR_RRCRelease(v_NGC_NASCellG);

    //@siclog "Step 68" siclog@

    v_AT_Response := f_UT_ReadAllowedNSSAI(UT);

    if (match (v_AT_Response, "")) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 68");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 68");

    }

  }

After change

function fl_TC_9_1_4_1_TestBody() runs on NR5GC_PTC

  {

    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_CellsOnDifferentPLMN);

    var NR_CellId_Type v_NGC_NASCellG := f_NR5GC_MapNASCell(ngc_CellG, NG_CellsOnDifferentPLMN);

    var NG_NAS_GutiParameters_Type v_GutiParamsCellA := f_NR5GC_CellInfo_GetGutiParameters(v_NGC_NASCellA);

    var NG_NAS_GutiParameters_Type v_NewGutiParams := v_GutiParamsCellA;

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var NG_5G_S_TMSI v_S_Tmsi :=  f_NR_Get5G_S_TMSI_FromGuti (v_GutiParamsCellA);

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;

    var NAS_PlmnId v_PLMN_CellA := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA));

    var NAS_PlmnId v_PLMN_CellG := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellG));

    var charstring v_Quote := oct2char ('22'O);

    var charstring v_FullNameString := "FullName12345678";

    var charstring v_ShortNameString := "SName123"; // @sic R5-198186 sic@

    var charstring v_Expression_CTZR := "*"&v_Quote&"(*)"&v_Quote&",([0-9])*"&v_Quote&"(*)"&v_Quote&"*"; // used for LTZ, DST and Time checking

    var charstring v_AT_Response;

    var PCCH_Config v_PcchConfig;

    var NSSAI v_RequestedNSSAI;

    var TypeOfRegistration_Type v_TypeOfRegistration;

    var integer i;

    var NR_RadioBearerId_Type v_SRB;
    var template (value) NSSAI v_ConfiguredNSSAI := {

      iei := '31'O,

      iel := '04'O,

      s_nssai := {cs_S_NSSAI(omit, '01'O, '02'O, omit), cs_S_NSSAI(omit, '01'O, '03'O, omit)}

      };

    var template (value) NSSAI v_AllowedNSSAI:= {

      iei := '15'O,

      iel := '06'O,

      s_nssai := {cs_S_NSSAI(omit, '01'O, '01'O, omit), cs_S_NSSAI(omit, '01'O, '02'O, omit), cs_S_NSSAI(omit, '01'O, '03'O, omit)}

      };

    timer t_T3555 := f_NR_SetTimerToleranceMin(v_NGC_NASCellA, nasTimer, 6.0);

    //@siclog "Step 1" siclog@

    v_S_Tmsi := f_CreateNewS_TMSING_5G_S_TMSI(v_S_Tmsi);

    v_NewGutiParams.AMF_SetId := substr(v_S_Tmsi, 0, 10);

    v_NewGutiParams.AMF_Pointer := substr(v_S_Tmsi, 10, 6);

    v_NewGutiParams.NG_TMSI := substr(v_S_Tmsi, 16, 32);

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB2,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(cs_ConfigUpdateInd('0'B, '1'B),

                                                                      f_NG_GutiParameters2MobileIdentity(v_PLMN_CellA, v_NewGutiParams)))));

    //@siclog "Step 2" siclog@

    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    //@siclog "Step 3" siclog@

    f_NR_RRCRelease(v_NGC_NASCellA);

    f_NR5GC_CellInfo_SetGutiParameters(v_NGC_NASCellA, v_NewGutiParams);

    v_PcchConfig := f_NR_GetSIB1_PcchConfig(v_NGC_NASCellA);

    //@siclog "Step 4" siclog@

    // Page with the S-TMSI used in the new GUTI

    f_NR_Paging(v_NGC_NASCellA,

                cs_PagingRecordList_OneRecord(cs_PagingUE_Identity_5G_S_TMSI(v_S_Tmsi)),

                v_PcchConfig.defaultPagingCycle); //f_NR_GetSIB1_PcchConfig(v_NGC_NASCellA).defaultPagingCycle;

    //@siclog "Steps 5-11" siclog@

    v_ReceivedMsg := f_NR5GC_RRCIdleToConnected_Steps2_4(v_NGC_NASCellA);

    f_NR5GC_RRC_ConnectedState3N_Steps5_8(v_NGC_NASCellA, v_ReceivedMsg);

    f_UT_AT_TriggerTimezoneReport (UT, "2010", "04", "12", "13", "??", "??", "\+04", "1");
    //@siclog "Step 12" siclog@

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB2,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(-, -, -, -, -,

                                                                      cs_NetworkName ('43'O, v_FullNameString),

                                                                      cs_NetworkName ('45'O, v_ShortNameString), // @sic R5-198186 sic@

                                                                      cs_TimeZone('40'O), // @sic R5-198186 sic@

                                                                      cs_TimeZoneAndTime('01'O, '40'O, '21'O, '31'O, '83'O, '25'O,'40'O),

                                                                      cs_DaylightSavingTime_tlv('01'B))))); // @sic R5-198186 sic@

    //@siclog "Steps 13" siclog@

    t_T3555.start;

    alt{

        [] SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)))

            {

                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");

            }

        [] t_T3555.timeout {}

    }

    //@siclog "Steps 14" siclog@

    v_AT_Response := f_UT_AT_CheckTimezoneReport (UT); // CTZE response expected

    f_AT_ResponseCheck(v_AT_Response,

                       "2010/04/12,13:[0-9]#(2):[0-9]#(2)", // TZ GMT+1, dst (1 char) ignored, check date and hour, ignore mins, secs 

                       "Step 14 (UT and local time zone)",

                       v_Expression_CTZR,

                       2);

    f_AT_ResponseCheck(v_AT_Response,

                       "1", // DST=1

                       "Step 14  DST",

                       v_Expression_CTZR,

                       1);

    v_AT_Response := f_UT_ChkOperatorName (UT, LONG_NAME);

    f_AT_ResponseCheck (v_AT_Response,

                        v_FullNameString,

                        "Step 14 (long network name)");

    v_AT_Response := f_UT_ChkOperatorName (UT, SHORT_NAME);

    f_AT_ResponseCheck (v_AT_Response,

                        v_ShortNameString,

                        "Step 14 (short network name)");

    v_AT_Response := f_UT_ChkTimeZone (UT);

    f_AT_ResponseCheck(v_AT_Response,

                       "*[0-9]#(2)/[0-9]#(2)/[0-9]#(2),[0-9]#(2):[0-9]#(2):[0-9]#(2)\+04*", // only TZ to be checked

                       "Step 14 (local time zone)");

     /* MOVED v_AT_Response := f_UT_AT_CheckTimezoneReport (UT); // CTZE response expected

    f_AT_ResponseCheck(v_AT_Response,

                       "2010/04/12,14:[0-9]#(2):[0-9]#(2)", // TZ GMT+1, dst (1 char) ignored, check date and hour, ignore mins, secs

                       "Step 14 (UT and local time zone)",

                       v_Expression_CTZR,

                       2);

    f_AT_ResponseCheck(v_AT_Response,

                       "1", // DST=1

                       "Step 14  DST",

                       v_Expression_CTZR,

                       1); */
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    //@siclog "Steps 15" siclog@

    // Switch off the UE

    f_NR_UE_DeRegisterOnSwitchOff (v_NGC_NASCellA, STATE_CONNECTED_3A);

    //@siclog "Step 16" siclog@

    // Switch on UE

    f_UT_SwitchOnUE(UT);

    //@siclog "Steps 17-19" siclog@

    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_Secure)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    //@siclog "Steps 20-30" siclog@

    f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo,

                                v_NGC_NASCellA, v_ReceivedMsg,

                                v_PLMN_CellA,

                                TESTMode_OFF,

                                Initial_Secure);

    //@siclog "Step 31" siclog@

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,

                                                   f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellA),

                                                   cs_NG_TAIListNonConsecutive(v_PLMN_CellA,

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(v_NGC_NASCellA))}),

                                                   -, cs_NSSAI_SST1eMBB ('15'O), // default value

                                                   -, v_ConfiguredNSSAI);

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Step 31A - 31E" siclog@

    f_NR5GC_RRC_Idle_Steps15_20(v_NGC_NASCellA,TESTMode_OFF,noRrcConnectionRelease); // @sic R5-198186 sic@

    //@siclog "Step 32A" siclog@ // @sic R5-198186 sic@

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB2,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(cs_ConfigUpdateInd('1'B, '1'B)))));

    //@siclog "Step 32A" siclog@

    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    f_NR_RRCRelease(v_NGC_NASCellA);

    //@siclog "Step 33" siclog@

    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected);

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility, -, -, -, -, -, -, ?)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 33");

    }

    // Need to check received contents (cannot hard-code a template to test this on a record of type)

    v_RequestedNSSAI.s_nssai := v_ReceivedMsg.Pdu.PiggybackedPduList[0].Msg.registration_Request.requestedNSSAI.s_nssai;

    for (i := 0; i < lengthof(v_RequestedNSSAI.s_nssai); i := i+1) {

      if (match (v_RequestedNSSAI.s_nssai[i].sd, v_ConfiguredNSSAI.s_nssai[0].sd) or match (v_RequestedNSSAI.s_nssai[i].sd, v_ConfiguredNSSAI.s_nssai[1].sd)) {

        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 33");

      } else {

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unknown Requested NSSAI");

      }

    }

    //@siclog "Step 34 - 42" siclog@

    // Void // @sic R5-198186 sic@

    //@siclog "Step 43" siclog@

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellA),

                                                   cs_NG_TAIListNonConsecutive(v_PLMN_CellA,

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(v_NGC_NASCellA))}),

                                                   -, v_AllowedNSSAI,

                                                   -,-,-,

                                                   v_GMM_MobilityInfo.SessionStatus //@sic R5-198509 sic@

                                                   );

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //@siclog "Step 44" siclog@

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

    }

    //@siclog "Step 44A" siclog@

    f_NR_RRCRelease(v_NGC_NASCellA); // @sic R5-198186 sic@

    //@siclog "Step 45" siclog@

    v_AT_Response := f_UT_ReadAllowedNSSAI(UT, f_ConvertPLMNtoString(v_PLMN_CellA));

    if (match (v_AT_Response, f_Convert_ListOfNSSAI_ForATCommand(v_AllowedNSSAI))) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 45");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 45");

    }

    //@siclog "Step 45A - 45H" siclog@

    // The test steps 1 to 8 of generic test procedure in TS 38.508-1 Table 4.5.4.2-3 are performed // @sic R5-198186 sic@

    f_NR5GC_RRC_ConnectedState3N_Def (v_NGC_NASCellA);

    //@siclog "Step 46" siclog@

    v_AllowedNSSAI.s_nssai := {cs_S_NSSAI(omit, '01'O, '01'O, omit), cs_S_NSSAI(omit, '01'O, '02'O, omit)};

    v_AllowedNSSAI.iel := '04'O;

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB2,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(cs_ConfigUpdateInd('0'B, '1'B), -, -, v_AllowedNSSAI))));

    //@siclog "Step 47" siclog@

    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    //@siclog "Step 48" siclog@

    v_AT_Response := f_UT_ReadAllowedNSSAI(UT, f_ConvertPLMNtoString(v_PLMN_CellA));

    if (match (v_AT_Response, f_Convert_ListOfNSSAI_ForATCommand(v_AllowedNSSAI))) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 48");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 48");

    }

    //SS releases the RRC connection.

    //@siclog "Step 48A" siclog@

    f_NR_RRCRelease(v_NGC_NASCellA);

    //@siclog "Step 49" siclog@

    f_NR_SetCellPower (v_NGC_NASCellG, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    f_NR_SetCellPower (v_NGC_NASCellA, tsc_NR_NonSuitableCellSSS_EPRE_FR1, tsc_NR_NonSuitableCellSSS_EPRE_FR2);

    //@siclog "Steps 50-52" siclog@

    v_TypeOfRegistration := f_NR5GC_InitialOrMobilityRegistrationRequest (v_GMM_MobilityInfo, v_ReceivedMsg, v_NGC_NASCellG);

    if (v_TypeOfRegistration == Initial_Secure) {

      // @siclog "Steps 53a1 - 53a10" siclog@

      f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo, v_NGC_NASCellG, v_ReceivedMsg, v_PLMN_CellG, TESTMode_OFF, v_TypeOfRegistration);

    }

    //@siclog "Step 53b" siclog@

    // do nothing if mobility

    //@siclog "Step 64" siclog@

    v_ConfiguredNSSAI.s_nssai := {cs_S_NSSAI(omit, '01'O, '02'O, omit), cs_S_NSSAI(omit, '01'O, '03'O, omit)};

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellG),

                                                   cs_NG_TAIListNonConsecutive(v_PLMN_CellG,

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(v_NGC_NASCellG))}),

                                                   -, cs_NSSAI_SST1eMBB ('15'O), // default value

                                                   -, v_ConfiguredNSSAI,

                                                   -,

                                                   v_GMM_MobilityInfo.SessionStatus //@sic R5-198509 sic@

                                                   );

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellG,

                                   tsc_NR_RbId_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    if (v_TypeOfRegistration == Initial_Secure) {

      // @siclog "Steps 65a1 - 65a5a1" siclog@

      f_NR5GC_RRC_Idle_Steps15_20 (v_NGC_NASCellG, TESTMode_OFF, noRrcConnectionRelease);

      v_SRB := tsc_NR_RbId_SRB2;
    } else {

      //@siclog "Step 65b" siclog@

      SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellG, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

      if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

      }

      v_SRB := tsc_NR_RbId_SRB1;

    }

    //@siclog "Step 66" siclog@

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellG,

                                   v_SRB,
                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND( cs_ConfigUpdateInd('0'B, '1'B), // @sic R5-198186 sic@

                                                                       -, -, -, -, -, -, -, -, -, -, -, cs_NetworkSlicingInd('0'B, '1'B)))));

    //@siclog "Step 67" siclog@

    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellG, v_SRB, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    //@siclog "Step 67A" siclog@

    f_NR_RRCRelease(v_NGC_NASCellG);

    //@siclog "Step 68" siclog@

    v_AT_Response := f_UT_ReadAllowedNSSAI(UT);

    if (match (v_AT_Response, "")) {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 68");

    } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 68");

    }

  }
MCC160 Implementation for change 4:
...  

    //@siclog "Steps 5-11" siclog@

    v_ReceivedMsg := f_NR5GC_RRCIdleToConnected_Steps2_4(v_NGC_NASCellA);

    f_NR5GC_RRC_ConnectedState3N_Steps5_8(v_NGC_NASCellA, v_ReceivedMsg);

    f_UT_AT_SubscribeTimeZoneReporting(UT);
    f_UT_AT_TriggerTimezoneReport (UT, "2010", "04", "12", "13", "??", "??", "\+04", "1");

    //@siclog "Step 12" siclog@

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,

                                   tsc_NR_RbId_SRB2,

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(-, -, -, -, -,

                                                                      cs_NetworkName ('43'O, v_FullNameString),

                                                                      cs_NetworkName ('45'O, v_ShortNameString), // @sic R5-198186 sic@

                                                                      cs_TimeZone('40'O), // @sic R5-198186 sic@

                                                                      cs_TimeZoneAndTime('01'O, '40'O, '21'O, '31'O, '83'O, '25'O,'40'O),

                                                                      cs_DaylightSavingTime_tlv('01'B))))); // @sic R5-198186 sic@

    //@siclog "Steps 13" siclog@

    t_T3555.start;

    alt{

        [] SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,

                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)))

            {

                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");

            }

        [] t_T3555.timeout {}

    }

    v_AT_Response := f_UT_AT_CheckTimezoneReport (UT); // CTZE response expected @sic R5s200573 sic@  //@siclog "Steps 14" siclog@

    f_AT_ResponseCheck(v_AT_Response,

                       "2010/04/12,13:[0-9]#(2):[0-9]#(2)", // TZ GMT+1, dst (1 char) ignored, check date and hour, ignore mins, secs

                       "Step 14 (UT and local time zone)",

                       v_Expression_CTZR,

                       2);

    f_AT_ResponseCheck(v_AT_Response,

                       "1", // DST=1

                       "Step 14  DST",

                       v_Expression_CTZR,

                       1);
    v_AT_Response := f_UT_ChkOperatorName (UT, LONG_NAME);

    f_AT_ResponseCheck (v_AT_Response,

                        v_FullNameString,

                        "Step 14 (long network name)");

    v_AT_Response := f_UT_ChkOperatorName (UT, SHORT_NAME);

    f_AT_ResponseCheck (v_AT_Response,

                        v_ShortNameString,

                        "Step 14 (short network name)");

    v_AT_Response := f_UT_ChkTimeZone (UT);

    f_AT_ResponseCheck(v_AT_Response,

                       "*[0-9]#(2)/[0-9]#(2)/[0-9]#(2),[0-9]#(2):[0-9]#(2):[0-9]#(2)\+04*", // only TZ to be checked

                       "Step 14 (local time zone)");












    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

...
2.3 Change 3

	Function name
	f_ConvertS_NSSAI_ForATCommand

	Reason for change
	Use of function oct2str can include a leading ‘0’ in the result string. To remove leading ‘0’ in result string, first convert oct to int and then convert int to string.

	Summary of change
	Call function int2str and function oct2int to get the correct result string.

	TTCN module
	NG_NAS_Common

	MCC160 Comment
	Rejected.  27.007 states that the S-NSSAI should be in hexadecimal format, not integer


Before 

function f_ConvertS_NSSAI_ForATCommand(template (value) S_NSSAI_Type p_S_NSSAI) return charstring

  {

    var S_NSSAI_Type v_S_NSSAI := valueof(p_S_NSSAI);

    var charstring v_S_NSSAIstring := "";

    v_S_NSSAIstring := oct2str(v_S_NSSAI.sst);

    if (ispresent(v_S_NSSAI.sd)) {

      v_S_NSSAIstring := v_S_NSSAIstring & "." & oct2str(v_S_NSSAI.sd);

    }

    if (ispresent(v_S_NSSAI.mappedSST)) {

      v_S_NSSAIstring := v_S_NSSAIstring & ";" & oct2str(v_S_NSSAI.mappedSST);

    }

    if (ispresent(v_S_NSSAI.mappedSD)) {

      v_S_NSSAIstring := v_S_NSSAIstring & "." & oct2str(v_S_NSSAI.mappedSD);

    }

    return v_S_NSSAIstring;

  }

After change

function f_ConvertS_NSSAI_ForATCommand(template (value) S_NSSAI_Type p_S_NSSAI) return charstring

  {

    var S_NSSAI_Type v_S_NSSAI := valueof(p_S_NSSAI);

    var charstring v_S_NSSAIstring := "";

    v_S_NSSAIstring := int2str(oct2int(v_S_NSSAI.sst));
    if (ispresent(v_S_NSSAI.sd)) {

      v_S_NSSAIstring := v_S_NSSAIstring & "." & oct2str(v_S_NSSAI.sd);

    }

    if (ispresent(v_S_NSSAI.mappedSST)) {

      v_S_NSSAIstring := v_S_NSSAIstring & ";" & oct2str(v_S_NSSAI.mappedSST);

    }

    if (ispresent(v_S_NSSAI.mappedSD)) {

      v_S_NSSAIstring := v_S_NSSAIstring & "." & oct2str(v_S_NSSAI.mappedSD);

    }

    return v_S_NSSAIstring;

  }
2.4 Change 4

	Function name
	f_UT_ReadNSSAI_DynamicParams

	Reason for change
	The use of regexp() in this function causes problems since the metacharacter * will match the longest possible number of characters. If V_Cnf.ResultString includes multiple “C5GNSSAIRDP” and the calling of regexp will filter not only the 1st “C5GNSSAIRDP” but also the subsequent “C5GNSSAIRDP”.

	Summary of change
	v_Cnf.ResultString can be returned directly

	TTCN module
	UpperTesterFunctions

	MCC160 Comment
	Accepted


Before 

function f_UT_ReadNSSAI_DynamicParams(UT_PTC_MTC_PORT p_Port,

                                      READ_NSSAI_PARAM_Type p_TypeToRead,

                                      template (omit) charstring p_PLMN_ID := omit) return charstring

  {

    var UT_COMMON_CNF v_Cnf;

    var charstring v_ReadType := int2str(enum2int(p_TypeToRead));

    if (isvalue(p_PLMN_ID)) {

      v_ReadType := v_ReadType & "," & valueof(p_PLMN_ID);

    }

    v_Cnf := fl_UT_ApplyCommandRetCnf(p_Port, cas_UT_Req(READ_NSSAI_PARAMS, CNF_REQUIRED, {cs_Ut_Parameter_Common ("NSSAIType", v_ReadType)}));

    return regexp(v_Cnf.ResultString, "*C5GNSSAIRDP: (*)", 0);
  }

After change

function f_UT_ReadNSSAI_DynamicParams(UT_PTC_MTC_PORT p_Port,

                                      READ_NSSAI_PARAM_Type p_TypeToRead,

                                      template (omit) charstring p_PLMN_ID := omit) return charstring

  {

    var UT_COMMON_CNF v_Cnf;

    var charstring v_ReadType := int2str(enum2int(p_TypeToRead));

    if (isvalue(p_PLMN_ID)) {

      v_ReadType := v_ReadType & "," & valueof(p_PLMN_ID);

    }

    v_Cnf := fl_UT_ApplyCommandRetCnf(p_Port, cas_UT_Req(READ_NSSAI_PARAMS, CNF_REQUIRED, {cs_Ut_Parameter_Common ("NSSAIType", v_ReadType)}));

    return v_Cnf.ResultString;
  }
2.5 Change 5

	Function name
	f_UT_ReadAllowedNSSA

	Reason for change
	1. ‘+’ is a regular expression metacharacter. Need to add escape charecter ‘\’ before it to interpret it a literal.
2. Metacharacter * will match the longest possible number of characters. To filter exactly what we want, add the matching of quotation marks around allowed NSSAI.

	Summary of change
	1. Add a metacharacter \ before +.
2. Add quotation marks matching around allowed NSSAI.

	TTCN module
	UpperTesterFunctions

	MCC160 Comment
	Accepted


Before 

function f_UT_ReadAllowedNSSAI(UT_PTC_MTC_PORT p_Port,

                                 template (omit) charstring p_PLMN_ID := omit,

                                 boolean p_Access3GPP := true) return charstring

  {

    var charstring v_ResultString := f_UT_ReadNSSAI_DynamicParams(p_Port, Default_Rejected_Configured_Allowed, p_PLMN_ID);

    var charstring v_Allowed := regexp(v_ResultString, "*<CR><LF>+C5GNSSAIRDP: ""*"",[0-9]+,""*"",(*)", 0);

    if (p_Access3GPP) {

      return regexp(v_Allowed, "[0-9]+,(""*"")", 0);

    } else {

      return regexp(v_Allowed, "[0-9]+,""*"",[0-9]+,(""*"")", 0);

    }

  }

After change

function f_UT_ReadAllowedNSSAI(UT_PTC_MTC_PORT p_Port,

                                 template (omit) charstring p_PLMN_ID := omit,

                                 boolean p_Access3GPP := true) return charstring

  {

    var charstring v_ResultString := f_UT_ReadNSSAI_DynamicParams(p_Port, Default_Rejected_Configured_Allowed, p_PLMN_ID);

    var charstring v_Allowed := regexp(v_ResultString, "*<CR><LF>\+C5GNSSAIRDP: ""*"",[0-9]+,""*"",(*)", 0);

    if (p_Access3GPP) {

      return regexp(v_Allowed, "[0-9]+,""(*)""*", 0);

    } else {

      return regexp(v_Allowed, "[0-9]+,""*"",[0-9]+,(""*"")", 0);

    }

  }
3 Execution Log Files 
3.1 HiSilicon Balong 5000  
The HiSilicon Balong 5000 5G UE passed this test case on Keysight 5G Protocol Conformance Toolset platform. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test case execution log file:

TC_9_1_4_1_LOG.html

In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

· PICS/PIXIT parameter file:
TC_9_1_4_1_PIXIT.xml

